
Til denne kogebog hører der nogle filer - færdige regneark og grafer. 
Foretrækker du at anvende Excel på din egen computer og ikke på skolens, så 
kommer du til at mangle disse filer. Din matematiklærer kan fortælle dig, 
hvorledes du får fat på disse filer. 

1. Fundamentale begreber 
 

Excel startes ved at klikke på Excel-ikonen:  
 
 

 
Efter et stykke tid kommer der et skærmbillede frem, som ligner nedenstående: 
 

 
Den fundamentale enhed i Excel er cellen. En celle kan enten indeholde et tal, en 
tekst eller en formel. 
 
En celle har en adresse, som består af søjlebogstavet og rækkenummeret -  
således befinder cellen A1 sig i den A'te søjle og den 1. række. 
 
Lad os indtaste tallet 123,45 i cellen B2. Dette gøres således: 
 
1) Flyt vha. musen den plus-formede 

markør hen til celle B2 og klik. B2 
omgives nu af en tyk, sort kant. 

 
2) Skriv 123,45. Skærmen skulle se ud 

 som følger: 
 
 
3) Man afslutter ved enten at trykke på ENTER-tasten eller ved at klikke på 

den lille firkant med det grønne rettetegn. Man kan afbryde indtastningen 
uden at ændre cellens indhold fra det oprindelige ved at taste ESC eller 
ved at klikke på det røde kryds.  

Bemærk, at decimaltegnet er et komma - ikke et punktum. På nogle maskiner er 
decimaltegnet dog et punktum. Hvis du er i tvivl, så spørg din lærer. 
 
Lad os indtaste teksten 'Hej med dig' i cellen C5: 
 

1) Klik på C5 og skriv 
 Haj med dig 
 
2) Tast ENTER 
  
3) Ups! - vi stavede  
 'Hej' forkert. Dette 
 rettes ved at klikke 
 C5. Man kan nu 
 rette  teksten  i 
  menubjælken. 
 

Lad os indtaste en formel i C8: 
 
Klik C8 og tast 
 =B2+47,4 
 
Denne formel tager indholdet i celle 
B2 og lægger 47,4 til. Resultatet 
vises i celle C8, mens man i 
editeringsbjælken kan se formlen. 
 
Excel er ligeglad med, om man taster 
B2 eller b2. 
 
En anden måde at indtaste denne 
formel er 
 

1) Klik C9. 
 
2) Tast  = 
 
3) Klik B2. Der kommer nu en 
 løbende, punkteret ramme 
 omkring cellen B2. Endvidere  
 står der B2 i editeringsbjælken. 
 
4) Tast   +47.4 
 
5) Tast ENTER. 

En blok er en rektangulær samling af celler. En blok betegnes ved celle1:celle2, 
hvor celle1 er celle-adressen på øverste venstre hjørne, og celle2 er adressen på 
nederste højre hjørne. 
 
Blokken A4:B11 kan markeres på følgende 
måde: Klik A4 og, mens museknappen holdes 
nede, træk musemarkøren ned til B11. 
Resultatet skulle blive som til højre: 
 
 
 
 
 

Søjler og rækker er specielle tilfælde af 
blokke. Man kan markere 8. række ved at 
klikke på 8-tallet i rækkenumrene til venstre i 
regnearket, og D-søjlen kan markeres ved at 
klikke på D'et øverst i regnearket. På figuren 
ses, hvordan Excel viser markering af en 
række: 

 

Tekst og tal kan formatteres. Prøv  

1) Klik C5 (cellen med teksten 'Hej med dig') 

2) Klik I-knappen i trykknappanelet. Teksten bliver nu kursiveret. Klikkes I-
knappen igen, så fjernes kursiveringen. 

3) Klik B-knappen. Denne gør teksten fed. 

4) Klikkes A-pilop eller A-pilned knapperne, så gøres teksten større eller 
mindre. 

5) Klikkes venstrestillingsknappen (lige til højre for A-pilned) placeres al 
tekst eller tal til venstre i cellen. Tilsvarende har man en centreringsknap 
og en højrestillingsknap. 

Flere formatteringsmuligheder kan fås ved at gå ind i menuen Format. Her kan 
man vælge: 

Alignment: Bestemmer bl.a., om teksten skal stå til venstre eller til højre eller 
midt i cellen. 

Font:  Bestemmer skrifttype, størrelse, fed/kursiv og tekstens farve. 

Border: Bestemmer typen og tykkelsen af rammen omkring cellen. 

Patterns: Bestemmer, om cellen skal skraveres. 

Number: Dette er nok det vigtigste menupunkt her. Dette bestemmer, 
hvorledes et tal skal vises - med antal cifre og decimaler og med 
eventuelle $-, kr.- £- og %-tegn. Den generelle regel er, at 

 
   # angiver et ciffer, som kan være der, 
   0 angiver et ciffer, som skal være der (også 0), 
   ? virker som et 0, bortset fra at ikke-betydende nuller 
      erstattes med et mellemrum 
   , er decimaltegnet. 
 
 Eksempelvis vil tallet 123.45 blive vist som følger: 
  0000,000  giver  0123,450 
  ####,###  giver  123,45 
  ####,000  giver  123,450 

Det anbefales at eksperimentere med disse faciliteter. Hjælp kan man altid få - 
man skal bare taste F1 eller vælge menupunktet Help.Contents eller, i enkelte 
dialogbokse, klikke på knappen Help. 
 
Format-menuen kan også kaldes frem ved at klikke på højre museknap. 

Øvelser 

1. Talformater 

Denne øvelse går ud på at eksperimentere med de forskellige talformater. 
Forsøg at skrive tallet til venstre på måden angivet til højre: 
 
a) 3 → 03   b) 3 → 3,00 
c)  3 → 03,00   d) 3,1415926 → 3,14 
e) 3,1415926 → 3,1415 f) 1000000 → 1,00E+06 
g) 7,35 → 7,35 kr.  h) 0,085 → 8,5% 

2. Sådan henter og gemmer man filer i Excel 
 
Når man skal gemme et regneark eller en graf i Excel, så gør man følgende: 
 
1) Vælg menuen File.Save As eller File.Save. Forskellen er, at man ved at 

bruge File.Save As kan ændre filnavnet, mens File.Save gemmer 
regnearket under det nuværende navn, som står øverst i regnearket. 

 
2) Ved File.Save gemmes filen bare uden videre dikkedarer. 
 
3) Ved File.Save As, som man altid bør benytte, når man har oprettet et nyt 

regneark, eller har ændret i et regneark, som andre også skulle kunne bruge 
senere i original tilstand, kommer nedenstående dialogbox frem: 

 

 
I rubrikken Drives kan man vælge mellem A-drevet (disketten) eller harddisken 
(C-drevet). Det er bedst at bruge A-drevet til private filer. I File Name kan man 
skrive det ønskede filnavn. Endelig kan man bladre rundt i Directories. 
 
Det er smart at vælge et sigende navn til sine filer - f.eks. er det en dårlig idé at 
kalde alle sine filer for REGNEARK.XLS. 
 
Et filnavn kan være på max. 8 bogstaver. Endvidere skal 'efternavnet' med - det er 
de tre bogstaver efter punktummet og afhænger af filtypen: 
 
 .XLS er til regneark 
 .XLC er til grafer (charts) 
 .XLW er til arbejdsmapper (workbooks) 
 .XLM er til makroer 
 .XLA er til en speciel type makroer (add-ins) 
 
De to sidste typer slipper man heldigvis for at bruge i det daglige. 

Når man vil hente en fil, så skal man benytte File.Open. Herved fremkommer 
følgende dialogbox: 
 
 

 
Dialogboxen virker på samme måde som ved gemning af filer - dog skal filnavnet 
enten skrives i rubrikken File Name, eller man kan vælge filnavnet ved at klikke 
det i fillisten nedenunder File Name. 
 
En arbejdsmappe (workbook) er en samling af regneark og grafer. Det er en 
ekstremt god idé at gemme sammenhørende regneark og grafer i samme 
arbejdsmappe. 
 
Man opretter en arbejdsmappe ved at kalde menuen File.New og vælge 
Workbook. Der kommer nu en indholdsfortegnelse (Workbook Contents), og 
man kan nu flytte eksisterende regneark eller grafer ind i arbejdsmappen ved at 
klikke trykknappen Add.. nederste til venstre. Nye regneark laves på samme måde. 
De tre knapper nederst til højre bruges til at bladre mellem arkene i 
arbejdsmappen. 
 
Mange af de filer, som anvendes i dette skrift, er gemt i arbejdsmapper. Disse 
ligger i biblioteket C:\EXCEL\KOGEBOG. 
 
Når man indlæser en af disse kogebogsfiler, så vil Excel bede dig at skrive et 
password, som du naturligvis ikke kender. Klik i stedet på knappen Read Only - 
dette betyder, at du ikke kan gennem kogebogs-filen under det navn, den har nu. 
Du kan stadigvæk ændre i filen eller gemme den under et andet navn, f.eks. på din 
private diskette.  

3. Parabler (eller absolut og relativ adressering) 
 
Formålet med denne øvelse er at tegne en parabel. Metoden kan i øvrigt bruges til 
tegning af graferne for andre funktioner. Alle regneark og grafer er gemt i 
arbejdsmappen PARABEL.XLW. 
 
1) Tag et rent (nyt) regneark. Dette kan f.eks. gøres ved at vælge File.New og 

vælge Sheet. 
 x-værdierne skal være i A-søjlen, og y-værdierne i B-søjlen. 
 
2) x-værdierne ordnes på følgende måde: 
 Klik A5 og tast  x  og  ENTER. 
 Klik A6 og tast  -10  og  ENTER. 
 Markér blokken A6:A26. 
 

  
 Vælg Data.Series. I dialogboxen, som vises ovenfor, vælges Stop Value 

til 10. 
 Klik OK - og observer, at hele blokken A6:A26 fyldes med tallene fra  
 -10 til 10. 
 
 Dette kan også gøres på en alternativ måde: 
 Tast -10 i celle A6 og -9 i celle A7. 
 Marker A6:A7. 

I den sorte rand omkring den markerede blok er der i 
nederste højre hjørne en lille sort firkant. Klik på 
denne (markøren, som normalt er et stort, hvidt plus, 
ændres til et lille, tyndt sort plus), og træk denne 
firkant med til celle A26. 

 Blokken A6:A26 fyldes med tallene. 
 
 (Eksperimentér iøvrigt med -10 i A6, -8 i A7 osv.). 
 
 
3) Vi vil tegne parablen med ligningen y x x= + −2 3 12 : 
 
 Indtast y i celle B5. 

 

Hér  

 I cellen B6 indtastes    =2*A6^2+3*A6-1      
 Efter at have tastet ENTER kommer den relevante y-værdi 169 frem i 

cellen - se figuren: 
 

   
 
 Klik B6 og træk det lille sorte hjørne ned til B26. Cellens indhold bliver 
 kopieret, og regnearket beregner alle y-koordinaterne. 
 
 Resultatet er gemt i regnearket PARABEL1.XLS 
 
 
4) Kig på B7's indhold: 
 

Det ses, at Excel ved 
kopieringen  fra B6 til B7 har 
ændret referencen til cellen A6 
til en reference til cellen A7. 
Dette er et eksempel på relativ adressering. 
 
Relativ adressering fungerer på følgende måde: Excel skal regne indholdet 
af celle B6 ud. Den støder på udtrykket A6 og tænker: 'Aha - cellen A6 er 
en celle til venstre for B6. Vi skal altså tage tallet lige til venstre og bruge 
dette tal i beregningerne'. 
 
Når B6's indhold bliver kopieret, så ændres celleadressen egentligt ikke - 
A7, som der nu refereres til, er stadigvæk cellen lige til venstre for den 
nuværende celle, nemlig B7. 
 
Modsætningen til relativ adressering er absolut adressering. Her 
henvises til en fast celle i regnearket, og denne henvisning ændres ikke ved 
kopiering eller lignende. Vil man f.eks. henvise til cellen A1 absolut, så 
skal man bruge dollar-tegn -  $A$1. Mere herom senere. 
 

5) Vi vil nu tegne parablen: 
 
  Marker A6:B26. 
  Vælg File.New og vælg Chart. 
  I den fremkomne dialogbox vælges X-values for XY-chart: 

 Der kommer nu en flot parabel. Der er bare et problem; punkterne er ikke 

forbundne med linier. (Grafen er gemt i PARABEL2.XLC) 
 
 Dette afhjælpes ved at vælge Gallery.XY(Scatter) og i dialogboxen at 

vælge mulighed nr. 2. (PARABEL3.XLC) 
 
6) Gå tilbage til regnearket. Dette kan f.eks. gøres ved at vælge 

Window.Sheet1 eller Window.Parabel1.xls (eller hvad regnearket nu 
hedder). 

 
7) Vi vil ændre regnearkets opbygning på følgende måde: Vi vil gerne kunne 

tegne parabler med ligningen y ax bx c= + +2 , men hvor vi let skal kunne 
ændre værdierne af parametrene a, b og c. Dette gøres ved at anbringe 
værdierne for a, b og c i cellerne B1, B2 og B3: 

 
  Tast  a=  i celle A1 
  Tast  b=  i celle A2 
  Tast  c=  i celle A3 
  Tast  2  i celle B1 
  Tast  3  i celle B2 
  Tast  -1  i celle B3 
  Tast  =B1*A6^2+B2*A6+B3  i celle B6 
  Kopiér indholdet af celle B6 med i blokken B6:B26. 
 
8) Det virkede ikke: Der kom masser af #'ere i cellerne. ØV! 
 Gør man B-søjlen lidt bredere (klik på stregen mellem B'et og C'et øverst 

og træk til højre) ser man, at der i B-cellerne står den noget kryptiske 
besked  

  #VALUE! 
 Dette er en fejlmeddelelse, og i dette tilfælde betyder det, at Excel ikke 

kan beregne indholdet af cellerne. 
  
 Lad os kigge lidt på regnearket: 

 I cellen B10 står der  
   =B5*A10^2+B6*A10+B7 
 Dette skyldes den relative adressering! 
 
 I cellen B6, som vi jo oprindeligt kopierede fra, stod der jo referencer til 

cellerne B1, B2 og B3. Dette læser Excel som 'cellen 5 oven over 
nuværende celle', 'cellen 4 oven over nuværende celle' og 'cellen 3 oven 
over den nuværende celle' - og kontrollerer man efter, så gælder dette 
stadig i celle B10. Men her refereres bl.a. til celle B5, som indeholder 
teksten 'y' - og hvad er en tekst gange et tal? 

 
9) Løsningen består i at bruge absolut adressering: 
 
  Tast  =$B$1*A6^2+$B$2*A6+$B$3  i celle B6 
  Kopiér dette ned til cellerne B6:B26. 
  
 Resultatet er gemt i  PARABEL4.XLS 
 
10) Prøv at ændre B1 (a-koefficienten) til -2. Alle tallene og grafen ændres 

med det samme. 
 
11) Vil man tegne flere parabler i samme koordinatsystem, så er det smart at 

gemme koefficienterne i søjlen over y-værdierne til den pågældende 
parabel. En ny parabel kan laves i C-søjlen som følger: 

 
  Tast  1  i cellerne C1, C2 og C3. 
  Tast  =B$1*$A6^2+B$2*$A6+B$3  i cellen B6 
  Kopiér B6's indhold ned til B6:B26. 
  Kopiér B6's indhold hen til B6:C6. 
  Kopiér C6's indhold ned til C6:C26. 

Her er der brugt blandet adressering: Referencen B$1 henviser absolut 
til 1.række, men relativt til B-søjlen - idet referencen er i B6, betyder 
dette 'cellen i 1.række og i samme søjle'. Tilsvarende betyder $A6 'cellen i 
samme række og i A-søjlen'. 

 
 Regnearket er gemt i PARABEL6.XLS 
 
13) Selve grafen kan nu laves ved 
  
  Markér A6:C26 
  File.New  Chart  X-values for XY-chart 
  Gallery.XY(Scatter) 2 
 
 Herved fremkommer der en graf med to parabler. (PARABEL7.XLC). 
 

Øvelser 

2. Annuitetslån 

Et annuitetslån er et lån, hvor man låner en vis sum penge til en fast rente. Hver 
termin (f.eks. hver måned) afbetaler man så et fast beløb, indtil man har afdraget 
hele lånet. 
 
Lad os sige, at vi låner 10000 kr. til en månedlig rente på 5%. Det månedlige 
afdrag er på 1000 kr. 
 
Efter 0 måneder skylder vi banken 10000 kr. 
Efter 1 måned skylder vi banken 10000 10000 0 05 1000 9500+ ⋅ − =,  kr. - de 
10000 0 05 500⋅ =,  kr. var jo den rente, banken tilskrev. 
Efter 2 måneder skylder vi 9500 9500 0 05 1000 8975+ ⋅ − =,  kr.  osv. 
 

 



13 

a) Lav et regneark, som viser, hvorledes det går med dette annuitetslån. I 
cellen A3 skal gælden efter 0 måneder stå, i cellen B1 skal gælden efter 1 
måned, men før der tilskrives rente eller afdrages (dette er lig med 
indholdet af A3). I cellen B2 skal den tilskrevne rente stå, og i cellen B3 
skal den resterende gæld efter 1 måned stå. 

 I cellerne C1, C2 og C3 skal der stå gæld før rente og afdrag, rente og 
gæld efter afdrag efter 2. måned. Osv. 

 Brug relativ adressering - det er lettest. 
 
b) Hov - du glemte vist, at der er tale om krone-beløb. Se så at få formateret 

alle tallene, således at alle tallene kun vises med to decimaler. 
 
c) Henne i banken vil man gerne kunne give alle kunder, som vil låne penge 

som annuitetslån, et regneark som ovenstående. Det skal altså vise, hvor 
meget man skylder, og hvor meget rente, der er tilskrevet, måned for 
måned. 

 Nu er det ikke alle kunder, som låner lige meget, og ydelsen kan også 
variere. Endelig er ikke alle kunder lige kreditværdige, så nogle må betale 
en højere rente end andre. 

 Lav et regneark, hvor lånebeløbet (grundstolen) indtastes i celle A1, 
ydelsen i celle A2, og renten i celle A3. Resten af regnearket skal minde 
om det regneark, du lavede i a). 

 (Vink: Brug absolut adressering til at hente grundstolen ($A$1), ydelsen 
($A$2) og renten ($A$3). 

 
d) Af hensyn til sin selvangivelse ønsker en kunde at få at vide, hvor meget 

hun kommer til at betale i rente i alt. Hvordan kan dit regneark fra c) løse 
dette problem? 

 (Vink: Excel-funktionen SUM(blok) lægger alle tallene sammen i den 
blok, som angives som 'indmad'.) 

3. Lagkagediagram 

Regnearket JORDEN1.XLS i arbejdsmappen JORDEN.XLW indeholder data 
over jordens grundstofsammensætning, samt over grundstofsammensætningen i 
lava og i sedimenter (aflejrede stenarter, f.eks. sandsten, kalk, etc...). 
 
Lav et lagkagediagram over grundstofsammensætningen i lava. 
 
 (JORDEN2.XLC indeholder det færdige resultat. Gallery.Pie 5 kan anbefales 
som graftype. Man kommer ud for at skulle vælge x-labels og y-values, som ikke 
hænger sammen i en blok. Dette kan gøres ved at vælge Chart.Edit series, klikke 
i x-labels, klikke hen til regnearket og markere grundstofferne i en stiplet, 
løbende ramme, og derefter gå ned til y-values, klikke hen til regnearket og igen 
markere de relevante værdier). 



14 

4. Lineær regression og bedste rette linie 
 
Her gives opskriften på, hvorledes man kan anvende Excel til at lave lineær 
regression, dvs. finde ligningen for den bedste rette linie. Endvidere vises, 
hvorledes denne bedste rette linie kan tegnes i en graf. 
 
Eksemplet, der anvendes, er Ohm's lov. Denne lov siger, at strømmen I gennem 
en resistor er proportional med U spændingsfaldet over resistoren: 
 U R I= ⋅ . 
R kaldes resistorens resistans. 
 
1) Tag et tomt regneark. I celle A1 indtastes I (Ampere), og i celle B1 

indtastes U (Volt). 
Søjlen A er for smal, så den kan gøres bredere ved at gå op mellem A'et og 
B'et og trække stregen mellem de to felter mod højre. 
 

2) I cellerne A2 og nedefter indtastes I-værdierne. De tilsvarende U-værdier 
indtastes i cellerne B2 og nedefter. (Se nedenfor) 

 
3) Tallene har ikke alle lige mange decimaler - det ser grimt ud. Gør derfor 

følgende: 
 Markér cellerne A1:B8. Vælg menupunktet Format.Number (eller klik på 

musens højre knap og vælg Number i menuen). I den fremkomne 
dialogbox skrives i feltet Code følgende '##0,##0'. Vælg OK. 

 Alle tallene skulle nu have 3 decimaler lige under hinanden. 
 
 Skærmen skulle nu se sådan ud: 
 
 Regnearket er gemt i (OMHSLOV1.XLS). 
 
 
 
 
 
 
 
4) Nu skal vi til at lave grafer: 
 Markér cellerne A2:B8. Vælg menupunktet File.New, og vælg Chart i den 

fremkomne dialogbox. 
 Der kommer nu en ny dialogbox. Klik ud for punktet X-values for XY-

Charts.  

 Vi ser nu følgende graf:  (OHMSLOV2.XLC) 
 

 
 
5) Prøv at gøre følgende: 
 Vælg menuen 
  Gallery.XY(Scatter) 
 og vælg nr. 2 ud af de mulige grafer. Vi får en graf, hvor målepunkterne er 

forbundne med liniestykker. Dette er vi egentligt ikke interesserede i her - 
vi vil hellere have den bedste rette linie, som ikke nødvendigvis falder 
sammen med alle punkterne. Vi skynder os derfor at vælge 
Gallery.XY(Scatter), mulighed nr. 1. Vi er nu tilbage i punkt 4). 

 

 
 
6) For at få Excel til at beregne den bedste rette linie, så skal vi lave en 

lineær regression. Dette gøres som følger: 

 Gå tilbage fra grafen til regnearket - klik enten på selve regnearket én 
gang, eller vælg menupunktet Window.Sheet1.XLS (eller måske 
Window.Ohmslov1.XLS). 

 
 I celle D1 skrives  Lineær regression, i celle D2 a= og i celle E2  b=. 
  
 Markér cellerne D3:E3 og skriv 
  
   =LINEST(B2:B8;A2:A8;TRUE;FALSE) 
 
 Afslut med at trykke CTRL-SHIFT-ENTER. (Dette er vigtigt). 
 

 
 I cellen D3 skrives nu hældningskoefficienten for den bedste rette linie 

(a-værdien), og skæringen med y-aksen (b-værdien) skrives i cellen E3. 
 
 Den bedste rette linie har altså ligningen 
 
   U I= ⋅ +9 911392 0 03667862, ,  
 
 og dette betyder, at resistorens resistans R er lig ca. 9,92Ω. 
 
 Dette regneark er gemt som OHMSLOV3.XLS. 
 
 Generelt virker LINEST-kommandoen på følgende måde: 
 

a) Man skal markere to celler ved siden af hinanden, som ovenfor. 
 b) Man skriver nu  =LINEST(  

c) Indmaden i LINEST kommer i fire portioner  
 - først skal man skrive den blok, hvor y-værdierne gemmer 
sig - dette er ovenfor I-værdierne i B2:B8. Så kommer der et 
semikolon, og dernæst x-værdierne - A2:A8 ovenfor. Igen et 
semikolon, og så enten et af ordene  TRUE  eller  FALSE.  Vælger 

man et  FALSE her, så sættes b automatisk til at være 0 - og det kan 
godt være lidt farligt. 
 Endelig kommer der et semikolon og igen enten  TRUE  
eller  FALSE. Vælg  FALSE  her, indtil du ved mere om lineær 
regression! 

 d) Afslut med Ctrl-Shift-ENTER! 
 
 
7) Vi vil nu indtegne den bedste rette linie på grafen, og her skal vi bruge C-

søjlen. For ikke at komme i karambolage med D-søjlen, som allerede er 
brugt, så laver vi en ekstra søjle: 

  
 Markér hele C-søjlen (klik øverst på C'et), og vælg menupunktet 

Edit.Insert. Der kommer nu en ekstra søjle, og D-søjlen omdøbes til E 
osv. 

 
8) I celle C1 skrives  Ureg (Volt)  - reg'er står for regression. I denne søjle 

skal de beregnede U-værdier stå. 
 I celle C2 skrives =TREND($B$2:$B$8 ; $A$2:$A$8 ; A2), og der 

trykkes på ENTER. 
 Markér celle C2, og klik på den lille sorte firkant i nederste højre hjørne. 

Træk nedad indtil cellerne C3:C8 er kommet med i den voksende, sorte 
kasse. 

 Slip museknappen, og cellerne C3:C8 fyldes med tal. Det skulle se sådan 
her ud: (OHMSLOV4.XLS) 

 

 
 Tallene i C-søjlen er y-værdierne for den bedste rette linie hørende til x-

værdierne i A-søjlen. 
 
 Generelt virker TREND på følgende måde: 
 
   TREND( kendte y-værdier  ;  kendte x-værdier ; x-værdi  ) 

 Den førdte indmadsdel er en blok af kendte y- værdier, og en blok af 

tilsvarende kendte x-værdier - det er ud fra disse, at regressionen 
foretages. Herefter kommer den x-værdi, for hvilken y-værdien skal 
beregnes. 

 
 $-tegnene i 1. og 2. del af indmaden er nødvendige, idet blok-adresserne 

for de kendte værdier ville ændre sig, når vi kopierede formlen. 
 
9) I C-søjlen står nu de tal, vi skal bruge for at lave bedste rette linie på 

grafen: 
 
 Markér cellerne A2:C8, vælg File.New.Chart, og vælg X-values for XY-

chart. 
 Vælg Gallery.XY(Scatter),  mulighed 2. 
 
 Klik på en af de små 'data-firkanter'. Herved kommer der nogle hvide, lidt 

større kasser inde midt på grafen: 
 

 
 Dobbeltklik på en af disse hvide firkanter. Der kommer en dialogbox, hvor 

du skal vælge Patterns yderst til højre. Der kommer en ny dialogbox: 

 I Line vælges Custom, og i rulleskakt-menuen Weight vælges mulighed 

nummer 2. I Marker vælges None. Vælg endelig OK. 
 
 Nu skulle den bedste rette linie sammen med de oprindelige målepunkter 

være der!. 
 
10) Endelig kan vi lave visse afpudsninger: 
 
 Vælg Chart.Attach text  Title  OK  og skriv Ohm's lov. 
 Vælg Chart.Attach text  x-axis  OK  og skriv I (Ampere). 
 Vælg Chart.Attach text  y-axis  OK  og skriv U (Volt). 
 Klik på en af data-firkanterne. I Marker vælg en hvid Background. og som 

Style vælges et kryds. Dette gør, at målepunkterne markeres med et kryds 
fremfor med en lille firkant. 

 
 Denne graf er gemt i (OHMSLOV5.XLC) og ser således ud 

 Vælg Chart.Gridlines  og vælg  Major  ved både x-aksen og y-aksen. 

Dette giver en millimeterpapirsagtig baggrund. (Prøv også Minor). 
  
 
 

Øvelser 

4. Elektromotorisk kraft 

Mens vi er i gang med ellæren, så kan vi ligeså godt undersøge nogle måledata 
omkring elektromotorisk kraft. Tager man et batteri, så afhænger dets 
polspænding U pol  af den strøm I, der tappes fra batteriet, på følgende måde: 
 U U R Ipol i= − ⋅0  
U0 kaldes batteriets tomgangsspænding eller elektromotoriske kraft, og Ri  er 
batteriets indre resistans. 
 
I regnearket EMK.XLS (for en gangs skyld ikke inde i en arbejdsmappe) er der 
angivet nogle målinger på et batteri. Brug disse målinger: 
 
a) Lav en graf over målingerne og indføj den bedste rette linie. 
 
b) Brug lineær regression til at finde U0 og Ri  for batteriet. 

 

5. Den synodiske måned 

En synodisk måned er den tid, der går mellem to fuldmåner, eller, hvad der er det 
samme, Månens omløbstid i sin rotation omkring Jorden. 
Idet Månen langsomt bremses op i sin rotation omkring Jorden, kan man forvente, 
at den synodiske måned ganske langsomt aftager gennem årmillionerne. 
En måde at måle, hvorledes den synodiske måned aftager, er ved at betragte 
forstenede muslinger. En muslingeskal er nemlig opdelt i perioder (svarende til 
en synodisk måned), som igen er opdelt i dag-vækstlag. Ved at optælle det 
gennemsnitlige dag-vækstlag i de forstenede muslinger kan man således finde ud 
af, hvorledes den synodiske måned har varieret gennem tiden. 
 
I regnearket SYNODE.XLS er indtastet sammenhørende værdier mellem 
 t  =  forsteningens alder i mill. år 
og 
 x  =  den synodiske måneds længde i dage. 
 
a) Lav en graf og vurder, om x afhænger lineært af t. 

b) Find konstanterne a og b i den lineære sammenhæng 
  x at b= +  
 
c) Nutidige astronomiske observationer viser, at den synodiske måneds 

længde aftager med 2 millisekunder pr. århundrede. 
 Er der overensstemmelse mellem dit svar i b) og disse astronomiske 

observationer? 
 

6. Enzym 

Man kommer engang i mellem ud for, at sammenhængen mellem to størrelser 
ikke er lineær; men at denne sammenhæng kan transformeres om til en lineær 
sammenhæng. Eksempelvis kan man undersøge et enzyms reaktionshastighed v 
som funktion af substratkoncentrationen c. (Et enzym er et protein, der virker 
som katalysator, typisk i en biologisk sammenhæng). Teoretiske  overvejelser 
antyder, at sammenhængen mellem v og c er af formen 

 
1 1
v

a
c

b= +  

 
altså en lineær sammenhæng mellem 1/c og 1/v. 
 
Regnearket ENZYM.XLS indeholder en række målinger af c og v. 
 
a) c-værdierne står i A-søjlen og v-værdierne i B-søjlen. Lav nogle formler, 

som gør, at  1/c-værdierne står i C-søjlen og 1/v-værdierne i D-søjlen. 
 
b) Lav en graf, som angiver, som den lineære sammenhæng ovenfor er 

opfyldt. Lav evt. bedste rette linie. 
 
c) Find konstanterne a og b. 

5. Logaritmisk regression 
 
Logaritmisk regression benyttes ved eksponentielle udviklinger. Ved en sådan 
udvikling har man en sammenhæng af formen 
 y b a x= ⋅  
mellem de variable x og y. En sådan sammenhæng giver iøvrigt en ret linie på 
enkeltlogaritmisk papir (ELK). 
 
I logaritmisk regression har man en række data, som man formoder hænger 
sammen via en eksponentiel udvikling. Problemet er da at bestemme 
koefficienterne a og b. 
 
1) I regnearket RADIO1.XLS  er der angivet en 

række tælletal fra en radioaktiv kilde. Disse tal 
ser ud som til højre. 

 
 Teorien bag radioaktivitet fortæller, at 

tælletallet (y) aftager eksponentielt med tiden 
(t). Formelen er 

   y b at= ⋅  
 Endvidere taler man om halveringstiden - 

denne er 

   T
a1 2

1 2
/

log( / )
log

=  

 
2) Vi starter med at indtegne målingerne i et ELK: 
 
  Markér blokken A2:B10 
  Vælg  File.New  Chart   X-values for XY-Chart 
  Vælg  Gallery.XY(Scatter) og vælg den 4. mulighed. 
  Klik på en af de små firkanter. Der kommer en Patterns-dialogbox 

 frem (se figuren). Vælg Marker til et kryds, og sæt Background til 
hvid.  

 Klik på y-aksen. Der kommer nedenstående dialogbox: 
 

 
 Vælg  den 3. mulighed i Axis.Weight. 
 Vælg  Scale  (knappen yderst til højre i dialogboxen). 
 

 Sæt  Minimum til 10 og Category(X) axis crosses at  til 10. 
 Klik  OK 

Vælg menupunktet Chart.Gridlines  
og vælg  både Major Gridlines  og  
Minor Gridlines  for x-aksen. 
Bemærk, at  Gridlines for y-aksen 
allerede er valgt. 
 
 
 
 
 
 
Klik den lodrette linie, som svarer til x-værdien 10. 
I den herved fremkomne dialogbox (Patterns) vælges den 3. mulighed ved 
Weight: 
 

 
 Gør det samme ved den vandrette linie svarende til y=100. 
 
 Disse to sidste ting gør det, at de store gitterlinier (Major Gridlines) 

bliver skarpere optrukket end de små. 
  
 Resultatet af alt dette er gemt i  RADIO2.XLC 
 
 
3) Nu skal vi til selve regressionen. 
 
 Start med at gå tilbage til regnearket. 

 Indtast 
  D1: Logaritmisk regression 
  D2: a= 
  E2: b= 
  Marker D3:E3 og indtast 
   =LOGEST(B2:B10 ; A2:A10 ; TRUE ; FALSE) 
  og afslut indtastningen med CTRL-SHIFT-ENTER 
 

 
 LOGEST-funktionen har samme syntaks som LINEST-funktionen. 
 Tallene i D3:E3 fortaller os, at 
  
  a = 0 964425.   og  b = 142 6063.  
  
 Vil man hellere have halveringstiden, så 
  F2: Halveringstiden er 
  F3: =log(0.5)/log(D3) 
 
 (Fordobingstiden kan fås ved i stedet at indtaste  =log(2)/log(D3) ) 
 
4) For at tegne bedste rette linie anvender vi funktionen GROWTH, som 
svarer til funktionen TREND. Det resulterende regneark er gemt i RADIO4.XLS 
 
  Markér C-søjlen og vælg  Edit.Insert. 
  C1: Tælletal(reg) 
  C2: =GROWTH($B$2:$B$10 ; $A$2:$A$10 ; A2) 
  Kopiér C2's indhold ned til C2:C10. 
 
5) Gå tilbage til grafen. Vælg Chart.Edit Series  og  vælg  New Series: 

 I Y-values skrives  =RADIO4.XLS!$C$2:$C$10  Vælg  OK.  
 (Generelt henviser adressen REGNEARK.XLS!BLOK  til blokken 

BLOK i regnearket  REGNEARK.XLS). 
Der kommer nu en række grønne firkanter. Klik på en af dem. Den 
velkendte Patterns-dialogbox kommer frem. Vælg  None  i  Marker,  
Automatic  i  Line  og sæt Color til at være sort. 

 Det endelige resultat er gemt i  RADIO5.XLC 

 
 
Øvelser: 

7. Befolkningstal 

I regnearket BEFOLK.XLS er angivet befolkningstallene for Danmark og 
Brasilien i perioden 1950-1990. 
 
a) Kan disse befolkningstal beskrives ved eksponentielle udviklinger? 
 
b) Angiv fordoblingstiden for  Brasiliens befolkningstal. 
 

8. Atmosfærens tryk 

Atmosfærens tryk aftager med højden. I regnearket TRYK.XLS er angivet værdier 
for luftens tryk p (i millibar) i højden h (i fod). 
 
Er der en lineær eller en eksponentiel sammenhæng mellem h og p? 
 
 

9. Logistisk vækst  

Størrelsen y vokser logistisk med tiden t, når der gælder sammenhængen 

 y
y

ke bt=
+

∞
−1

 

 
Her er k, b og y∞  konstanter. Især størrelsen y∞  er vigtig - den angiver den 
maksimale værdi, y kan antage. 
 
Logistisk vækst optræder mange steder i naturen - typisk dér, hvor der er tale om 
en hæmmet vækst. 
 
a) Omskriv ligningen for den logistiske vækst til 

  
y
y

ke bt∞ −− =1  

 Hvilken sammenhæng er der mellem  t og 
y
y
∞ − 1 ? 

 
Man kan altså eftervise logistisk vækst ved at indtegne sammenhørende værdier af 

t og 
y
y
∞ − 1 på enkeltlogaritmisk papir og lave enkeltlogaritmisk regression. 

Problemet er at bestemme konstanten y∞  før denne regression. 
 
Regnearket SOLSIKKE.XLS indeholder data for sammenhængen mellem højden 
af solsikkeplanter og deres alder.  
 
b) Lav en almindelig graf over disse data og bestem y∞  ud fra denne graf. 
 
c) Lav den enkeltlogaritmiske graf og bestem konstanterne b og k. 
 
Regnearket FÆROERNE.XLS indeholder data over Færøernes landbefolkning 
(dvs. den del af befolkningen, som bor udenfor Torshavn).  
 
d) Lav en tilsvarende analyse af Færøernes landbefolkning. Vokser denne 

også logistisk? 
 
 

10. Kemisk termodynamik 

For en kemisk reaktion afhænger ligevægtskonstanten K af temperaturn T (i 
Kelvin) på følgende måde 

 ln K
H

RT
S

R
= − +

∆ ∆
 

hvor R er gaskonstanten, ∆H  reaktions entalpi og ∆S  reaktionens entropi. 
 
I regnearket NO.XLS er vist sammenhørende værdier for K og T for reaktionen 

 N O 2NO2 2+ →  
 
a) Brug disse data til at lave en (1/T , lnK)-graf. Er denne en ret linie? 
 
b) Bestem ∆H  og ∆S  for reaktionen. 

6. Dobbeltlogaritmisk regression 
 
Ved dobbeltlogaritmisk regression undersøger man sammenhænge af formen 
 y b x a= ⋅  
altså potentielle udviklinger. 
Dette kan Excel ikke umiddelbart klare, men ved et simpelt trick kan det alligevel 
lade sig gøre - tager man logaritmen af ligningen ovenfor, så får man 
 log log logy b a x= + ⋅ . 
Det er altså nok at lave lineær regression på log x  og log y. 
 
Af praktiske årsager er det bedst at bruge den naturlige logaritme - ln . (Dette 
skyldes, at Excel har indbygget den omvendte funktion til ln - den hedder exp - 
men ikke den omvendte til 10-tals-logaritmen log.) 
 
1) I regnearket PLANET1.XLS i arbejdsmappen PLANET.XLW er der for 

alle solsystemest planeter indtastet afstanden R til Solen (målt i 
astronomiske enheder - A.U.) og omløbstiden T i år. 

 

 
 Keplers 3.lov siger, at 
   T k R= ⋅ 3 2/  
 hvor k er en for sammenhængen uinteressant konstant. 
 
 Dette et velegnet til dobbeltlogaritmisk regression - tallet a er her 1.5. 
 
2) Vi starter med at tegne en graf: 
 
  Marker B2:C10 
  File.New.Chart  X-values for XY-chart 
  Gallery.XY(Scatter)  mulighed nr. 5 

 Denne graf er tegnet i et dobbeltlogaritmisk koordinatsystem (DOLK) og 
er gemt i PLANET2.XLC 

 
3) Vi laver nu selve regressionen: 
 
  F1: Dobbeltlogaritmisk regression 
  F2: a= 
  G2: ln(b)= 
  H2: b= 
  Markér F3:G3 og skriv 
   =LINEST(LN(C2:C10) ; LN(B2:B10) ) 
   CTRL-SHIFT-ENTER 
  H3: =EXP(G3) 
 
 I cellen F3 står der nu, at a=1.500151, hvilket er i strålende 

overensstemmelse med Kepler's 3. lov. 
 
 I LINEST-kommandoen gjorde vi det, at vi tog den naturlige logaritme af 

alle dataerne i x- og i y-søjlerne.  
 
4) Bedste 'rette' linie beregnes nu. Igen bruger vi lineær regression - i dette 

tilfælde TREND - på logaritmerne af dataerne. Vi skal dog nuske på, at 
outputtet af TREND også er en logaritme, og at vi derfor skal tage EXP af 
dette output: 

 
 D1: T(reg) 
 D2: =EXP(TREND(LN($C$2:$C$10);LN($B$2:$B$10);LN(B2))) 
 Kopiér D2's indhold ned til D2:D10. 
  
 Dette regneark er gemt i PLANET4.XLS 

5) Gå tilbage til grafen. 
 Vælg  Chart.Edit Series.   Vælg New Series og sæt  Y-values til 
 PLANET4.XLS!$D$2:$D$10 
 Klik en af de grønne firkanter, og sæt Line til Automatic, Color til sort og 

Marker til None. 
 Klik en af de røde firkanter, sæt Style til et kryds og Background  til hvid. 
 
 Grafen er gemt i PLANET5.XLC 
 
 
 

Øvelser 
 

11. Iltforbrug 

Regnearket ILT.XLS indeholder data for sammenhængen mellem kropsmasse M 
(målt i kg) og iltforbrug  V (liter O2/time) for nogle forskellige pattedyr. 
 
a) Vis, at der er en potentiel sammenhæng mellem M og V. 
 
b) En hval vejer 10 tons. Hvor stort er mon dens iltforbrug? 
 
 

12. Pærer 

Sammenhængen mellem elektriske pærers effektforbrug P og deres lysstryke L 
er angivet i regnearket PÆRE.XLS. (P måles i Watt, L i lumen). 
 
a) Lav en graf og forklar, at der er tale om en potentiel udvikling: 
  L b Pa= ⋅  
 
b) Bestem a og b. 

7. Statistik 
 
Vi viser her, hvorledes EXCEL kan anvendes til forskellige statistiske 
undersøgelser. 
 
Regnearket KLASSE7.XLS indeholder en liste over en (fiktiv) gymnasieklasse 
med 29 elever. ( KLASSE8.XLS indeholder nogenlunde de samme data, men skal 
bruges i en helt anden sammenhæng i næste kapitel). 
 
A-søjlen indeholder elevernes navne, B-søjlen informationer om elevernes køn. 
Herefter kommer deres telefonnumre, deres årsindkomster (af hensyn til 
skattevæsenet) og endelig deres karakterer i en række fag: Skriftlig dansk (DAs), 
mundtlig dansk (DAm), engelsk (EN), tysk (TY), skr. matematik (MAs), mdt. 
matematik (MAm), fysik (FY), religion (RE), geografi (GE), musik (MU). 
Endelig kommer der to karakterer i elevernes valgfag - den skriftlige (HNs) og 
den mundtlige (HNm). Underligt nok har alle eleverne valgt højniveaufag, og i 
den sidste søjle står der, hvilket fag dette er (KE=kemi, MU=musik, LA=latin, 
BK=billedkunst, BI=biologi, SA=samfundsfag). 
 
Karaktererne er nok noget lave - bl.a. forekommer der adskillige 00 og 03'ere. 
(Man får Excel til at skrive det første 0 ved at vælge Format.Number og sætte 
Code lig '00'.) 
 
1. Den enkelte elevs gennemsnit findes på følgende måde: 
   Skriv Gennemsnit i celle R1 
   I R2 skrives  =AVERAGE(E2:P2) 
   Kopiér dette udtryk ned til blokken R3:R30 

  Markér R2:R30 og vælg talformatet '0.0' (Gennemsnit 
beregnes kun med et decimal.) 

 
EXCEL indeholder mange statistiske funktioner, som virker på samme måde som 
AVERAGE. Her er en liste 
 
 AVERAGE( blok )  beregner gennemsnittet af tallene i blok 

MODE( blok )  finder typetallet i blok (Typetallet er det 
 hyppigst forekommende tal) 

MEDIAN( blok )  finder medianen ('det midterste' tal). 
COUNT( blok )  finder antallet af tal i blok 
SUM( blok )  finder summen af alle tallene i blok 
STDEV( blok )  standardafvigelsen 
VAR( blok )  variansen 
MIN( blok )  minimum af tallene i blok 
MAX( blok )  maksimum af tallene i blok 

 
Lidt mere specielle er 

QUARTILE( blok ,0)  minimum af tallene i blok 
QUARTILE( blok ,1)  1. kvartil af tallene i blok 
QUARTILE( blok ,2)  2. kvartil af tallene i blok (=medianen) 
QUARTILE( blok ,3)  3. kvartil af tallene i blok 
QUARTILE( blok ,4)  maksimum af tallene i blok 
 

 
2. Nu vil man gerne kigge de enkelte lærere lidt nærmere i kortene - er der 

f.eks. en sammenhæng mellem dansklærerens mundtlige og skriftlige 
karakterer (der skulle der gerne være...) 

 Dette kan undersøges ved at beregne 
   =CORREL(E2:E30 ; F2:F30) 
 som giver 0,717775... 
 CORREL( blok1 ; blok2 ) beregner korrelationskoefficienten mellem 

tallene i blok1 og blok2. Dette er et tal mellem 1 og -1 - jo nærmere 1, 
denne korrelationskoefficient er, jo bedre hænger de to blokkes tal 
sammen. 

 Vi beregnede korrelationen mellem E2:E30, som indeholdt de skriftlige 
danskkarakterer, og F2:F30, som indeholdt de mundtlige, og vi fik en 
korrelation på 0,72, hvilket er ret godt. 

 Korrelationen mellem de to højniveau-karakterer er ikke helt god: 
   =CORREL(O2:O30 ; P2:P30) 
 gav 0,002165.... En korrelationskoefficient, som er tæt på 0 tyder på, at 

der ingen sammenhæng er mellem to talsæt. Højniveau-lærerne har måske 
en svag tendens til at give karakterer helt hen i skoven. 

 Endelig kan vi se, at klassens matematiklærer er lidt sær: 
   =CORREL(I2:I30 ; J2:J30 ) 
 giver -0,126247... En negativ korrelation betyder, at jo højere den ene 

karakter er, jo lavere er den anden. 
 
3. Klassens fysik-lærer skal i forbindelse med en forældrekonsultation lave 

et histogram over klassens fysik-karakterer. Hvordan gør han det? 
 - Jo han bruger funktionen FREQUENCY ! 
 
  Skriv i celle AA1: Karakterer i fysik 
    AA2: Karakter 
    AB2: Hyppighed 
    AA3: 00 
    AA4: 03 
    AA5: 5 
    AA6: 6 
    AA7: 7 
    AA8 8 
    AA9: 9 
    AA10: 10 
    AA11: 11 

    AA12: 13 
 Markér blokken AB3:AB13 og skriv 
    =FREQUENCY( K2:K30 ; AA3:AA12 ) 
 Afslut med  Ctrl-SHIFT-ENTER ! 
 
 Frekvenserne (eller rettere hyppighederne) kommer nu i cellerne 

AB3:AB13. 
 
  Generelt virker funktionen FREQUENCY som LINEST eller 

LOGEST. Syntaksen er 
   =FREQUENCY( datablok ; fordelingsblok ). 
 Dataene ligger i datablok, mens fordelingsblok er en liste over de tal, 

hvor man opdeler dataintervallerne. F.eks. giver en fordelingsblok på 
tallene 0 ; 3.5 ; 7.5 ; 12  en opdeling af dataene efter intervallerne 

   −∞ ∞; ; ; . ; . ; . ; . ; ; ;0 0 3 5 3 5 7 5 7 5 12 12  
 
4. Histogrammet kan nu laves: 
   Markér AA3:AB12 
  Vælg  File.New  Chart og vælg muligheden Category (X) 

Axis Labels i den dialogbox, der fremkommer. 
 
 
5. Rektor vil af overskuelighedsårsager gerne have alle 'dumpe-karaktererne' 

5, 03 og 00 markeret med rødt. Hvordan gør han det? 
 
 I nogle af EXCEL's standard-talformater indgår der semikoloner (;). F.eks. 

er der et format, som hedder 
   ###0,00 ; [Red]###0,00 ; 0  
 Positive tal dækkes ind af den første trediedel af formatet (indtil ; ) - de 

bliver altså vist som ####0,00 (med 2 decimaler). Negative tal står i 
midten - de bliver også vist som ###0,00, men nu i rød skrift. Endelig 
bliver nullet vist som et 0 (den sidste trediedel). 

 
 Løsningen på rektors problem er at ændre formatet på karaktererne til 
   [<5][Red]00 ; [>5]#0 ; [Red] #0 
 [<5] betyder nu, at hvis tallet i en celle er mindre end 5, så skal det skrives 

med rødt og med nullet foran. [>5] betyder, at alle tal større end 5 skal 
skrives med 1-2 cifre i almindeligt farvet skrift, mens resten (som jo er 5-
tallerne), skal skrives med rød skrift. 

 
  Prøv at forudsige, hvad følgende formatering vil gøre ved 

karakterlisten 
   [<=5] [Red] 00 ; [>=10] [Blue] #0 ; #0 
 og se, om det passer. 

Øvelse 
 

13. Gymnasieklasse 

Samtlige delopgaver nedenunder drejer sig om regnearket KLASSE7.XLS 
 
a) Bestem gennemsnittet af samtlige karakterer indenfor hvert fag. 
 Hvilken lærer er mest fedtet med karaktererne? 
 
b) Skatteinspektøren vil gerne have at vide, hvor meget hele klassen tjener 

om året. Hvordan beregner han dette? 
 
c) Religions- og geografilæreren er faktisk én og samme person. Kan man se 

det på hendes karaktergivning? 
 
d) Den lidt halvskøre matematiklærer skal give hele klassen en telefonisk 

besked. Han får det indfald at ringe til gennemsnittet af hele klassens 
telefonnumre. Hvordan beregner han det? 

 
e) Hvilken elev har det højeste karaktergennemsnit? og det laveste? 
 
f) Man kunne tro, at elever, som tjener mange penge, har formentligt masser 

af erhvervsarbejde i fritiden, og de har derfor ikke tid til at læse lektier 
derhjemme. Er der faktisk en sammenhæng mellem karaktergennemsnit og 
indkomst? 

8. Databaser 
 
Vi vil nu fortsætte vores undersøgelse af gymnasieklassen fra før. Denne gang 
bruger vi regnearket KLASSE8.XLS. Dette minder meget om KLASSE7.XLS, 
men der er indføjet en søjle med elevernes fødselsdatoer, og en søjle med 
elevernes karaktergennemsnit. 
 
Vi vil opfatte alla dataene som en database: 
 
En database består af en hel masse poster (records) - i dette tilfælde er der en 
post for hver elev. Hver post indeholder nogle felter (fields), f.eks. navn, 
fødselsdato, karaktergennemsnit etc. etc. 
 
1. Først skal man have fortalt Excel, at man opfatter dataene som en database. 

Dette gøres ved at: 
   Markér A1:S31 
   Vælg  Data.Set Database 
 Teknisk set sker der kun det, at Excel nu kalder blokken A1:S31 for 

Database. 
 Man skal tage feltnavnene med, her A-rækken, og man bør lade en tom 

række være med også - det er en fordel, når man skal indsætte nye poster. 
 
2. En database kan sorteres. Vil vi f.eks. sortere eleverne efter alder, så 
   Markér A2:S30 
   Vælg Data.Sort.. 
  I den fremkomne dialogbox skal man nu vælge 1-3 feltnavne samt 

bestemme, om posterne skal sorteres voksende (Ascending) eller 
aftagende (Descending). Sorteringen sker leksikografisk, dvs. først 
sorteres efter første nøgle (1. Key), og indenfor poster, hvor den første 
nøgle er end, sorteres efter anden nøgle. 
  Vi vælger 1. Key som Født - klik på C1 (eller en anden celle 

i C-søjlen). 
 Klik OK. 
Eleverne sorteres nu efter alder (eller rettere fødselsdato). 
 

3. Nye poster kan oprettes, eller gamle poster slettes eller rettes, på to 
måder - den enkelste er ganske simpelt at rette direkte i regnearket. 

 
  En post oprettes ved at indsætte en ny række (markér nederste, 

tomme række i databasen og vælg Edit.Insert). Herved kommer der en tom 
række, som man kan udfylde. Bemærk, at man selv skal sørge for, at 
formelfelter som f.eks. karaktergennemsnittet bliver udfyldt. 

 
 En post slettes ved at markere den række, posten befinder sig i, og vælge 

Edit.Delete. 

4. Man kan også kalde Data.Form, hvorved der fremkommer en dialogbox 
kaldet en formular (form), som er lettere at styre. 

 Man bladrer mellem databaserne ved at klikke Find Next eller Find Prev 
eller ved at køre op eller ned i elevatorskakten.  Man sletter den 
nuværende post ved at klikke Delete, og man kan indtaste en ny post ved at 
klikke New. Endelig kan man rette i en post ved ganske simpelt gå ind og 
rette de felter, der skal rettes. 

 Bemærk, at man ikke kan komme til at pille ved feltet Gennemsnit - dette 
skyldes jo, at dette er et beregnet felt, som afhænger af alle de andre 
felter. 

 
5. En af de vigtigste ting, man kan lave med en database, er at søge. Når man 

søger i en database, så skal man have nogle kriterier, som skal være opfyldt 
hos de poster, man søger efter. 

 
 Det er nemmest at søge vha. formularen fra før. Man klikker bare Criteria 

og indtaster de kriterier, som er relevante. Ved at klikke Find Next eller 
Find Prev kan Excel nu finde de poster, som opfylder alle kriterierne 
samtidigt. 

 
 Et kriterium kan være af formen 
   >tal  <tal  >=tal  <=tal 

 =tal 
 for tal, datoer etc. 
   text  >text  <text  >=text 

 <=text 
 for tekst. Indtaster man som tekstkriterie An, så finder Excel alle de 

poster, hvor teksten starter ned An, f.eks. både Anne, Anders, Andreas, 
Andeby (V)And(L)and etc. Er man kun interesseret i teksten 'An', så skal 
man taste 

   ='=An' 
 
6. I Excel kan man ostå opstille kriterierne i en såkaldt Criteria Range. 

Denne fremgangsmåde tillader meget mere fleksible datasøgninger. 
 En Criteria Range laves som følger: 
   Marker A1:S1 
   Vælg Edit.Copy 
   Klik AA1 
   Vælg Edit.Paste 
 Nu skulle alle feltnavnene pænt stå i cellerne AA1:AS1. 
   Markér AA1:AT2 
   Vælg Data.Set Criteria 
 Blokken AA1:AT2 indeholder nu kriterierne. 
 Cellerne AT1:AT2 skal vi bruge senere, skriv f.eks. XX i celle AT1.  

 Man får også brug for en Extract Range, som er en blok indeholde de 
poster, som opfylder kriterierne. Denne laves ved 

   Markér A1:S2 
   Edit.Copy 
   Klik AA8 
   Edit.Paste 
   Markér AA8:AS36 
   Data.Set Extract 
 
7. Kriterierne indtastes ganske som ved formularen. Man kan dog godt have 

formler med 
   = DAm >= RE eller  =RE+TY-EN 
 altså de poster, hvor den mundtlige danskkarakter er mindst lige så stor 

som religionskarakteren. 
 Kriterier, som indtastes i samme række skal alle være opfyldt (logisk 'og'). 

Vil man have flere kriterier, hvor mindst ét skal være opfyldt (logisk 
'eller'), så skal man skrive kriterierne i forskellige rækker. 

 
 Ved at vælge Data.Extract fyldes Extract Range med de poster, som 

opfylder kriterierne. 
 
8. For at finde de piger, som er født efter 1/1-1977, tjener mindre end 

50000,- om året og er bedre til tysk end til engelsk, skriver vi  
   AB2  Pige 
   AC2  >1/1/1977 
   AE2  <50000 
   AT2  =TY>EN 
 og vælger Data.Extract. Nede i Extract Range kommer nu de poster, som 

opfylder kriterierne. 
 
 I AT2 står der et udsagn (TY>EN). Excel udvælger alle de poster, for 

hvilket dette udsagn er sandt. 
    
 
9. Hvem er bedst til matematik, drenge eller piger? 
   U40  Skr 
   U41  Mdt 
   V38  Køn 
   V39  Dreng 
   W38  Køn 
   W39  Pige 
 V38:V39 og W38:W39 indeholder nu to små Criteria Ranges, som 

vælger henholdsvis drengene eller pigerne. 



38 

   V40  =DAVERAGE( Database ; J1 ; V38:V39 ) 
   V41  =DAVERAGE( Database ; K1 ; V38:V39 ) 
   W40  =DAVERAGE( Database ; J1 ; W38:W39 ) 
   W41  =DAVERAGE( Database ; K1 ; W38:W39 ) 
 
 Som det ses, er pigerne bedre til skriftlig matematik (7.7 versus 6.7), 

mens drengene er bedre til det mundtlige (8.6 versus 8.3). 
 
 DAVERAGE er et eksempel på en database-funktion. Der findes flere: 
 
   DCOUNT  tæller antallet af poster 
   DSUM beregner summen 
   DMAX finder maximum 
   DMIN  finder minimum 
   DPRODUCT beregner produktet 
   DVAR beregner variansen 
   DSTDEV beregner spredningen 
 
 Alle funktionerne virker på følgende måde 
  
   =DAVERAGE ( database ; felt ; kriterier ). 
 
 Excel løber alle posterne i database igennem, udvælger de poster, der 

opfylder kriterier, og beregner gennemsnittet (summen 
/produktet/variansen...) af feltet felt. 

 
 Da vi oprettede databasen, gjorde kommandoen Data.Set Database jo det, 

at vi kaldte blokken med all posterne for 'Database'. Det er derfor nok at 
skrive Database som det første argument. Men man kan også vælge en 
anden blok, f.eks. hvis man kun vil bruge nogel af databaserne. 

 
 Kriterie-blokken kan være lige så stor, som man vil have det. 

9. Normalfordeling 
 
Excel kan ikke lave sandsynlighedspapir (eller probitdiagrammer eller 
normalfordelingspapir...). Men man kan alligevel komme langt. 
 
I arbejdsmappen NORMALFD.XLW indeholder regnearket NORMAL1.XLS 
følgende data: I 1883 målte man højden af mænd født i Wales, Irland, Skotland og 
England. Højderne er angivet i engelske tommer, og de observerede hyppigheder 
er angivet under hvert fødeland. 
 
1. Vil vi finde middelværdierne og spredningerne for de 4 måleserier. Vi kan 

ikke benytte Excel-funktionerne AVERAGE og STDEV, fordi vi ikke har 
selve målingerne, men kun hyppighederne. Men vi kan gøre følgende (se 
NORMAL2.XLS): 

 
   A24:  Ialt 
   A25:  Middelværdi  
   A26:  Spredning 
   B24:  =SUM(B2:B22) 
   B25: 

 =SUMPRODUCT($A$2:$A$22;B2:B22)/B24 
   B26: 

 =SQRT((B24*SUMPRODUCT($A$2:$A$22; 
     $A$2$A$22;B2:B22)-SUMPRODUCT( 
     $A$2:$A$22;B2:B22)^2)/(B24*(B24-1))) 
   ($-tegnene er kun nødvendige, fordi vil vil kopiere 

formlerne) 
   Kopiér B24:B26 til hele blokken B24:E26. 
 
2. For at kunne lave probitdiagrammet skal vi finde de kumulerede 

frekvenser. En måde at gøre dette på er først at finde de kululerede 
hyppigheder: 

 
   G1:  Wales, kum 
   H1:  Irland, kum 
   I1:  England, kum 
   J1:  Skotland, kum 
   G2:  =SUM(B$2:B2) 
   Kopiér G2 ned til blokken G2:G22, og kopiér G2:G22 til 

G2:J22. 
 
 Bemærk den kreative anvendelse af $-tegnet. For at finde de kumulerede 

frekvenser skal vi blot dividere den kumulerede hyppighed med antallet af 
observationer: 

 
   L1:  Wales, kf 

   (M1, N1, O1 tilsvarende overskrifter) 
   L2:  =G2/B$24 
   Kopiér L2 ud over blokken L2:O22. 
 
 For at kunne lave probitdiagrammet mangler vi blot at transformere de 

kumulerede frekevnser via den inverse normalfordeling. 
 
   Q1:  Wales, probit 
   (do. for R1, S1 og T1) 
   Q2:  =NORMSINV(L2) 
   Kopiér Q2 ud over Q2:T22. 
 
 Lav nu et Scatter(XY)-diagram med A2:A22 som x-værdier og Q2:Q22 

som y-værdier. Nogle af y-værdierne er ikke definerede - dette skyldes, at 
vi beder Excel om at beregne den inverse normalfordeling af 0 og 1, og det 
er umuligt. Disse værdier skal nok udelades fra y-værdi-serien. 

 
 Vi får en flot ret linie, så højden af waliserne er normalfordelt - se 

NORMAL3.XLS og WALES.XLC. 
 
 


